

# **Introduction to Lambda Calculus through redex**

By Jacob J. A. Koot

10/06/2009 00:23:10

## Introduction

Every Scheme programmer knows the word ‘lambda’, which is the name of the Greek letter ‘ $\lambda$ ’. Not every Scheme programmer knows the beautiful field of mathematics from which the use of the word ‘lambda’ originates. This piece of mathematics is called ‘Lambda Calculus’.<sup>(1)</sup> Knowledge of the Lambda Calculus is not a prerequisite in order to become a good programmer. However, for a Scheme programmer with only some basic mathematical knowledge and some curiosity to know the ideas on which Scheme and other functional programming languages are built, it is not difficult to grasp the basics of the mathematical theory. In this essay I attempt to provide a short and simple introduction into the Lambda Calculus to the aforementioned audience. See [Barendregt](#) for a comprehensive treatment of the Lambda Calculus.

## Formal languages

The Lambda Calculus is a formal mathematical system that, very much like a programming language, has a grammar and a set of semantic rules. The grammar tells us which forms a program can take. The semantics tell us what a program means, in particular how a program (interpreted or compiled and run) produces its results. The Lambda Calculus can be regarded as the ultimate mathematical abstraction of all programming languages. It consists of abstraction (lambda) and application only, nothing else, no numbers, no characters, no lists, no strings, no vectors and not even one single primitive procedure. Therefore the Lambda Calculus is not suited for real life programming. Nevertheless, the Lambda Calculus allows the description of every feasible computation. For example, one can describe Scheme and a universal Turing machine in terms of the Lambda Calculus. It will be shown how to represent data such as numbers and how to create self reference and recursion without any of the self referential Scheme forms such as *define*, named *let* and *letrec*. Because the Lambda Calculus is a mathematical theory, one can reason about it with mathematical methods. Therefore the Lambda Calculus is an important tool in scientific studies on the properties of real life programming languages.

## Tools and source texts

This essay comes with a set of files providing tools and all source texts used and shown in this essay. They depend on [PLT Scheme](#), particularly on PLT’s library [redex](#). See [appendix A](#) for instructions how to download and use PLT Scheme and the source texts. It takes a few mouse clicks only. Make sure to use a copy of this essay downloaded and stored as described the appendix.

## Reasoning *within* or *about* a formal system<sup>(3)</sup>

The grammar of the Lambda Calculus tells us which forms a program can take. A correctly composed program is called a ‘lambda term’ or a ‘well formed sentence’. The semantics consist of the rules of reduction and tell us how a program is evaluated. In order to find the properties and the results of a language or one of its programs, one can slavishly follow the rules of reduction just like an interpreter would do. This can be called: “reasoning *within* the system”. However, one may step out of the system and deduce properties of a language by contemplating the properties of its grammar and its semantics. This can be called: “reasoning *about* the system”. The MIU puzzle provides a good illustration of the difference between the two approaches. The language of this puzzle consists of all (proper) lists consisting of the symbols M, I and U.<sup>(4)</sup> The language and its rules are:

1 There are several Lambda Calculi. Here  $\lambda K\beta$  and  $\lambda K\beta\eta$  are treated and are simply referred to as *the* Lambda Calculus.

2 H. P. Barendregt, The Lambda Calculus, Its Syntax and Semantics, North-Holland, ISBN 0-444-87508-5.

3 With thanks to [Douglas R. Hofstadter](#) who permitted me to borrow and adapt the MIU puzzle and to borrow the distinction between *within* and *about* from his book “Gödel, Escher, Bach, an Eternal Golden Braid”. This book is a must!

4 In order to avoid an abundance of quotes, data or programs or fragments thereof, are implicitly quoted where necessary.

$\langle \text{sentence} \rangle ::= (\langle \text{symbol} \rangle \dots)$   
 $\langle \text{symbol} \rangle ::= M \mid I \mid U$

- 0  $(M I)$  is given.  
 1  $(\langle \text{symbol} \rangle \dots I)$   $\Rightarrow (\langle \text{symbol} \rangle \dots I U)$   
 2  $(M \langle \text{symbol} \rangle \dots)$   $\Rightarrow (M \langle \text{symbol} \rangle \dots \langle \text{symbol} \rangle \dots)$   
 3  $(\langle \text{symbol} \rangle_0 \dots I I I \langle \text{symbol} \rangle_1 \dots)$   $\Rightarrow (\langle \text{symbol} \rangle_0 \dots U \langle \text{symbol} \rangle_1 \dots)$

We may regard  $(M I)$  as an axiom. Every sentence that can be deduced from the axiom by means of zero or more applications of rules 1, 2 and 3 can be regarded as a theorem, the axiom included. The left hand side of a rule is a pattern to be matched. The right hand side a template. Each rule indicates that if a sentence matching the pattern is a theorem, the template is a theorem as well. Every sequence ' $\langle \text{symbol} \rangle \dots$ ' matches an arbitrary sequence of the symbols  $M$ ,  $I$  and  $U$ , but within an application of a rule every occurrence of ' $\langle \text{symbol} \rangle \dots$ ' represents the same sequence. Subscripts are used where it is necessary to distinguish between possibly different sequences (rule 3). Rule 1 states that symbol  $U$  may be added at the end of a sentence that ends with  $I$ . Rule 2 indicates that a sentence starting with  $M$  may be replaced by the sentence formed by appending all that follows the starting  $M$ , at the end of the original sentence. Rule 3 indicates that three consecutive symbols  $I$  may be replaced by one single symbol  $U$ . The question is: "Is  $(M U)$  a theorem?" The puzzle can easily be described in PLT's *redex*:

```
#lang scheme (require redex) ; File miu-puzzle.ss
(provide MIU-rules traces)

(define-language MIU-language ; Grammar.
  (<sentence> (<symbol> ...))
  (<symbol> M I U))

(define MIU-rules ; Semantics.
  (reduction-relation MIU-language
    (--> (<symbol> ... I) (<symbol> ... I U) "rule 1")
    (--> (M <symbol> ...) (M <symbol> ... <symbol> ...) "rule 2")
    (--> (<symbol>_0 ... I I I <symbol>_2 ...) (<symbol>_0 ... U <symbol>_1 ...) "rule 3")))
```

In *redex* a subscript is written as an underscore followed by a natural number, id est an exact non negative integer number. In order to see part of the graph of deductions starting from the axiom, try:

```
#lang scheme ; File miu-traces.ss
(require "MIU-puzzle.ss")
(traces MIU-rules '(M I))
```

Take a look at the graphical user interface opened by procedure *traces* and play with it. Procedure *traces* comes from PLT's library *redex*. It forms a graph of all theorems that can be deduced from axiom  $(M I)$ . However, the procedure is protected such as to halt when the graph outgrows some maximum size. Click the reduce button in order to find more theorems. Notice that many theorems can be deduced in more than one way. If proposition  $(M U)$  is a theorem, then we may repeatedly press the reduce button until list  $(M U)$  appears in the graph. If proposition  $(M U)$  is a theorem, procedure *traces* ultimately will find it. If  $(M U)$  is not a theorem, it will never be found, of course, regardless of the number of times the graph is extended by means of the reduce button. The number of different theorems that can be deduced from the axiom is infinite. For example, the second rule implies that from axiom  $(M I)$  we can deduce every theorem  $(M I \dots)$  whose number of occurrences of symbol  $I$  is a power of two. If the number of different theorems deducible from  $(M I)$  would be finite, then procedure *traces* eventually would signal that no more theorems can be found.

### Question

Is  $(M U)$  a theorem? Think about it before turning to the next page.

**Answer**

$(M\ U)$  is not a theorem. If the graph of deductions would be finite, this could be checked by inspecting the whole graph and look whether or not  $(M\ U)$  appears in the graph. However the graph is infinite. In order to prove that  $(M\ U)$  is not a theorem, we must step out of the system and look at the rules from a higher point of view. Now the proof is easily found. The number of occurrences of symbol  $\perp$  in axiom  $(M\ \perp)$  is not a multiple of 3. This property is conserved by the three rules of deduction. Hence every theorem has the property. Because proposition  $(M\ U)$  does not have the property, it is not a theorem.

When studying the Lambda Calculus, it is important to be aware in which mode we are working: reasoning *within* the system or reasoning *about* the system.

**The grammar of the Lambda Calculus**

In Backus-Naur notation the grammar of the Lambda Calculus as used in this essay is: <sup>(5)</sup>

$\langle \text{term} \rangle ::= \langle \text{var} \rangle$	Variable reference
$\langle \text{term} \rangle ::= (\lambda (\langle \text{var} \rangle) \langle \text{term} \rangle)$	Abstraction
$\langle \text{term} \rangle ::= (\langle \text{term} \rangle \langle \text{term} \rangle)$	Application
$\langle \text{var} \rangle ::= \text{identifier, but not } \lambda$	We don't allow $\lambda$ to be used as the name of a variable.

That's all. There are no semantic restrictions. Notice that in PLT Scheme  $\lambda$  is a synonym of `lambda`. In this essay  $\lambda$  is used. Free, id est unbound, variables will sometimes be used. In Scheme they are not permitted or at least give rise to problems when referenced. However, one might imagine a Scheme dialect in which the value of an unbound variable is the very same symbol that names the variable. There are, or at least have been, some Lisp dialects with this property. In fact it is not difficult to implement such a language for PLT Scheme, for example by a module that exports all of Scheme but renames `top` as `datum`. See files [free-vars-in-scheme.ss](#) and [test-free-vars-in-scheme.ss](#). Because the Lambda Calculus has no `define`-forms, there are no globally bound variables. When talking about lambda terms it is handy to have names for its components:

$\langle \text{var} \rangle$  is a subterm, particularly a variable reference, when occurring as an instantiation of a  $\langle \text{term} \rangle$ .

In  $(\lambda (\langle \text{var} \rangle) \langle \text{term} \rangle)$  the  $\langle \text{term} \rangle$  is a subterm and is called the body of the abstraction

In  $(\lambda (\langle \text{var} \rangle) \langle \text{term} \rangle)$  the  $\langle \text{var} \rangle$  is a formal argument. This occurrence of the  $\langle \text{var} \rangle$  is not a subterm.

In  $(\langle \text{term} \rangle_0 \langle \text{term} \rangle_1)$   $\langle \text{term} \rangle_0$  is a subterm, particularly the operator of an application.

In  $(\langle \text{term} \rangle_0 \langle \text{term} \rangle_1)$   $\langle \text{term} \rangle_1$  is a subterm, particularly the actual argument of an application.

The above grammar is easily rewritten for *redex*. In addition we prepare some metapredicates and auxiliary non terminals like variable lists and environments. Later they will appear to be useful.

```
#lang scheme ; File curried-lc-grammar.ss
(require redex)
(include "define-language-with-metapredicates.ss")
(provide curried-lc-grammar Term? Abstr? Appl? Var? Varlist? Env? Binding?)

(define-language-with-metapredicates curried-lc-grammar
  ( $\langle \text{term} \rangle$        $\langle \text{var} \rangle$   $\langle \text{abstr} \rangle$   $\langle \text{appl} \rangle$  #:pred Term?      )
  ( $\langle \text{abstr} \rangle$      $(\lambda (\langle \text{var} \rangle) \langle \text{term} \rangle)$  #:pred Abstr?    )
  ( $\langle \text{appl} \rangle$       $(\langle \text{term} \rangle \langle \text{term} \rangle)$       #:pred Appl?     )
  ( $\langle \text{var} \rangle$        $(\text{variable-except } \lambda)$  #:pred Var?      )
```

5 The conventional notation (without shorthand) for the Lambda Calculus is:

$\langle \text{term} \rangle ::= \langle \text{var} \rangle$  Use variables that do not require separators, for example arbitrary frames of pixels, except ' $\lambda$ '.

$\langle \text{term} \rangle ::= (\lambda \langle \text{var} \rangle \langle \text{term} \rangle)$  Abstraction

$\langle \text{term} \rangle ::= (\langle \text{term} \rangle \langle \text{term} \rangle)$  Application

For example:  $((\lambda x(xx))(\lambda x(xx)))$  which is called ' $\Omega$ '.

We use a notation that allows lambda terms more easily to be processed by Scheme for tests and demonstrations.

; Additional forms.

```
(⟨varlist⟩    (⟨var⟩ ...)           #:pred Varlist? )
(⟨env⟩        (⟨binding⟩ ...)       #:pred Env?      )
(⟨binding⟩    (⟨var⟩ number)        #:pred Binding? )
(⟨bool⟩       #f #t                 )
(⟨subterm⟩    (λ (⟨var⟩) ⟨subterm⟩) (⟨subterm⟩ ⟨term⟩) (⟨term⟩ ⟨subterm⟩) hole))
```

Within the context of *curried-lc-grammar*, the identifiers that appear as the first one in a clause, like `⟨term⟩` and `⟨var⟩`, are non terminals. To obtain a well formed lambda term, start with `⟨term⟩` and apply the rules of the grammar one or more times such as to obtain a sentence without non terminals. *Redex* provides many tools to check the consistency of its use, for example:

```
#lang scheme ; File test-curried-lc-grammar.ss
(require redex "curried-lc-grammar.ss")
(redex-check curried-lc-grammar ⟨term⟩
  (or ; Redex-check randomly forms 10000 terms and checks that this condition is satisfied.
    (term (Var?  ⟨term⟩)) ; Here we check that every ⟨term⟩ is either a variable reference,
    (term (Abstr? ⟨term⟩)) ; abstraction or
    (term (Appl?  ⟨term⟩))) ; application.
  #:attempts 10000 #:retries 100)
```

## Scope

Apart from occasional free variables and the absence of global bindings, the Lambda Calculus has the same rules of scope as Scheme (actually rather reversely). The formal argument of an abstraction is bound within the body of that abstraction and bindings may be shadowed. In the example below  $x_1$  creates a binding.  $x_2$  creates a binding that shadows that of  $x_1$ .  $x_3$  is a variable reference related to  $x_2$ .  $x_4$  is a free occurrence of variable  $x$ .

```
((λ (x) (λ (x) x)) x)
  1    2 3 4
```

## Free variables

Metafunction *Free-vars* finds the list of free variables of a lambda term. The produced list does not contain duplicates. Metafunction *Var-free-in?* returns `#t` or `#f` depending on whether or not a given variable occurs free in a given term.

```
#lang scheme ; File free-vars.ss
(require redex "curried-lc-grammar.ss")
(provide Free-vars Var-free-in?)

(define-metafunction curried-lc-grammar Free-vars : ⟨term⟩ -> (⟨var⟩ ...)
  ((Free-vars ⟨var⟩) (⟨var⟩))
  ((Free-vars (⟨term⟩_0 ⟨term⟩_1)) (Union (Free-vars ⟨term⟩_0) (Free-vars ⟨term⟩_1)))
  ((Free-vars (λ (⟨var⟩) ⟨term⟩)) (Remove-var ⟨var⟩ (Free-vars ⟨term⟩))))

(define-metafunction curried-lc-grammar Var-free-in? : ⟨var⟩ ⟨term⟩ -> ⟨bool⟩
  ((Var-free-in? ⟨var⟩_0 ⟨var⟩_0) #t)
  ((Var-free-in? ⟨var⟩_0 ⟨var⟩_1) #f)
  ((Var-free-in? ⟨var⟩_0 (λ (⟨var⟩_0) ⟨term⟩)) #f)
  ((Var-free-in? ⟨var⟩_0 (λ (⟨var⟩_1) ⟨term⟩)) (Var-free-in? ⟨var⟩_0 ⟨term⟩))
  ((Var-free-in? ⟨var⟩ (⟨term⟩_0 ⟨term⟩_1))
    , (or (term (Var-free-in? ⟨var⟩ ⟨term⟩_0))
          (term (Var-free-in? ⟨var⟩ ⟨term⟩_1)))))
```

```
(define-metafunction curried-lc-grammar Union : (<var> ...) (<var> ...) -> (<var> ...)
  ((Union (<var>_0 ...) ()) (<var>_0 ...))
  ((Union () (<var>_1 ...)) (<var>_1 ...))
  ((Union (<var>_0 <var>_1 ...) (<var>_2 ... <var>_0 <var>_3 ...))
    (<var>_0 ,@(term (Union (<var>_1 ...) (<var>_2 ... <var>_3 ...))))
  ((Union (<var>_2 ... <var>_0 <var>_3 ...) (<var>_0 <var>_1 ...))
    (<var>_0 ,@(term (Union (<var>_1 ...) (<var>_2 ... <var>_3 ...))))
  ((Union (<var>_0 <var>_1 ...) (<var>_2 <var>_3 ...))
    (<var>_0 <var>_2 ,@(term (Union (<var>_1 ...) (<var>_3 ...))))

(define-metafunction curried-lc-grammar Remove-var : <var> (<var> ...) -> (<var> ...)
  ((Remove-var <var>_0 (<var>_1 ... <var>_0 <var>_2 ...) (<var>_1 ... <var>_2 ...))
  ((Remove-var <var> (<var>_0 ...)) (<var>_0 ...))
```

The clauses in a definition of a metafunction have the form  $\langle \text{pattern} \rangle \langle \text{template} \rangle$ , where the pattern must always begin with the name of the metafunction and the template is implicitly wrapped as in  $(\text{term } \langle \text{template} \rangle)$ . Syntax *term* is like a quasiquotation, allowing *unquote* and *unquote-splicing*, but also recognizing calls to metafunctions and the bindings introduced in the pattern.

```
#lang scheme ; File test-free-vars.ss
(require "free-vars.ss" redex)

(test-equal (term (Free-vars ( $\lambda$  (x) (x x)))) (term ( )))
(test-equal (term (Free-vars ( $\lambda$  (x) (x y)))) (term (y)))
(test-equal (term (Free-vars ( $\lambda$  (x) ((x y) z)))) (term (z y)))
(test-equal (term (Free-vars (( $\lambda$  (x) (x y)) ( $\lambda$  (y) (x y)))) (term (y x)))
(test-equal (term (Var-free-in? x ( $\lambda$  (y) z))) #f)
(test-equal (term (Var-free-in? x ( $\lambda$  (x) x))) #f)
(test-equal (term (Var-free-in? x ( $\lambda$  (y) x))) #t)

(test-results) ; Displays: All 7 tests passed.
```

## $\alpha$ -congruence

Consider the lambda terms:  $(\lambda (x) x)$  and  $(\lambda (y) y)$ . When interpreted as representations of the (untyped) identity function, both terms represent the very same function. The two terms are said to be  $\alpha$ -congruent. The  $\alpha$ -congruence of two terms can be decided by a metafunction as shown in the next piece of code. We use environments in order to keep track of the bindings of variables.

```
#lang scheme ; File alpha-congruence.ss
(require redex "curried-lc-grammar.ss")
(provide  $\alpha$ -congruent?)

(define-metafunction curried-lc-grammar  $\alpha$ -congruent? : <term> <term> -> <bool>
  (( $\alpha$ -congruent? <term>_1 <term>_2) ( $\alpha$ -aux? (<term>_1 ()) (<term>_2 ( ))))

(define new-scope ; Use semaphore because the guis of traces may run concurrently.
  (let ((semaphore (make-semaphore 1))) (scope 0))
  ( $\lambda$  ()
    (begin0 (begin (semaphore-wait semaphore) (set! scope (add1 scope)) scope)
      (semaphore-post semaphore))))
```

```

(define-metafunction curried-lc-grammar  $\alpha$ -aux? : (<term> <env>) (<term> <env>) -> <bool>
  (( $\alpha$ -aux? (<var>_0 <env>_0) (<var>_1 <env>_1))
  , (let ((e0 (assq (term <var>_0) (term <env>_0))) (e1 (assq (term <var>_1) (term <env>_1))))
      (or (and e0 e1 (= (cadr e0) (cadr e1)))
          (and (not e0) (not e1) (eq? (term <var>_0) (term <var>_1))))))
  (( $\alpha$ -aux? (( $\lambda$  (<var>_0) <term>_0) <env>_0) (( $\lambda$  (<var>_1) <term>_1) <env>_1))
  , (term-let ((g (new-scope)))
      (term ( $\alpha$ -aux? (<term>_0 ((<var>_0 g) , @ (term <env>_0)))
                    (<term>_1 ((<var>_1 g) , @ (term <env>_1))))))
  (( $\alpha$ -aux? (((<term>_0 <term>_1) <env>_0) (((<term>_2 <term>_3) <env>_2))
  , (and (term ( $\alpha$ -aux? (<term>_0 <env>_0) (<term>_2 <env>_2)))
        (term ( $\alpha$ -aux? (<term>_1 <env>_0) (<term>_3 <env>_2))))))
  (( $\alpha$ -aux? any_1 any_2) #f))

```

```

#lang scheme ; File test-alpha-congruence.ss
(require redex "alpha-congruence.ss")

(define-syntax tester ; Does concurrent tests.
  (syntax-rules ()
    ((tester <term> expected)
      (void
        (set! nr-of-tests (add1 nr-of-tests))
        (thread (lambda () (test-equal (term <term>) expected) (semaphore-post sema))))))

(define nr-of-tests 0)
(define sema (make-semaphore 0))

(tester
  ( $\alpha$ -congruent?
    (( $\lambda$  (x) ( $\lambda$  (y) ( $\lambda$  (z) ((x y) z)))) a)
    (( $\lambda$  (p) ( $\lambda$  (q) ( $\lambda$  (r) ((p q) r)))) a))
  #t) ; Because there is a one to one correspondence between all variables.

(tester
  ( $\alpha$ -congruent?
    (( $\lambda$  (x) ( $\lambda$  (y) ( $\lambda$  (z) ((x y) z)))) a)
    (( $\lambda$  (x) ( $\lambda$  (y) ( $\lambda$  (z) ((x y) z)))) b))
  #f) ; Because free variable a  $\neq$  free variable b.

(tester
  ( $\alpha$ -congruent?
    ( $\lambda$  (x) ( $\lambda$  (y) ( $\lambda$  (z) ((z z) z))))
    ( $\lambda$  (x) ( $\lambda$  (x) ( $\lambda$  (x) ((x x) x))))
  #t) ; Because the underlined abstractions have no references to the other two bindings.

(tester
  ( $\alpha$ -congruent?
    ( $\lambda$  (x) ( $\lambda$  (y) ( $\lambda$  (z) ((z z) z))))
    ( $\lambda$  (x) ( $\lambda$  (z) ( $\lambda$  (y) ((z z) z))))
  #f) ; Because z has different binding levels.

(let loop ((nr-of-tests nr-of-tests)) ; Wait until after the threads have finished before
  (if (zero? nr-of-tests) (test-results) ; checking the test results.
      (begin (semaphore-wait sema)
              (loop (sub1 nr-of-tests))))) ; Displays: All 4 tests passed.

```



## Curry

Although the observation was first made by M. Schönfinkel, the name of H. B. Curry is attached to the idea that every function of a fixed number of one or more arguments can be written in terms of functions of one single argument. Applications must be adapted accordingly. For example:

Uncurried	Curried
$(\lambda (x\ y)\ x)$	$(\lambda (x)\ (\lambda (y)\ x))$
$(a\ b\ c\ d)$	$((a\ b)\ c)\ d$
$((\lambda (x\ y)\ x)\ p\ q)$	$((\lambda (x)\ (\lambda (y)\ x))\ p)\ q$

In order to improve readability, uncurried notation and superfluous parentheses will be allowed, but should always be regarded as shorthand for a fully curried form. This applies to both abstractions and applications. To indicate this, we write  $\langle\text{uncurried-term}\rangle \equiv \langle\text{corresponding-curried-term}\rangle$ .

```
#lang scheme ; File uncurried-lc-grammar.ss
(require redex "curried-lc-grammar.ss")
(provide uncurried-lc-grammar)

(define-extended-language uncurried-lc-grammar curried-lc-grammar
  (<term> <var> ( $\lambda$  (<var> ...) <term>) (<term> <term> ...)))
```

The new language is like the original one, but with the definition of the non terminal  $\langle\text{term}\rangle$  replaced by the new one. The process of transforming abstractions and applications of more than one argument to nested abstractions and applications of one argument is called ‘currying’. The uncurried shorthand of application is left associative, id est,  $(a\ b\ c)$  means  $((a\ b)\ c)$ , not  $(a\ (b\ c))$ . As a Schemer you already know that these two expressions usually have different behaviour. The formal arguments of an abstraction in uncurried shorthand are not required to be distinct. A formal argument may shadow a formal argument of the same name to its left. For example:  $(\lambda (x\ y\ x)\ z) \equiv (\lambda (x)\ (\lambda (y)\ (\lambda (x)\ z)))$ . We shall need a metafunction that transforms an uncurried or partially uncurried lambda term into a fully curried one.

```
#lang scheme ; File curry.ss
(require redex "uncurried-lc-grammar.ss")
(provide Curry)

(define-metafunction uncurried-lc-grammar Curry : <term> -> <term>
  ((Curry ( $\lambda$  () <term>)) (Curry <term>))
  ((Curry ( $\lambda$  (<var>_0 <var>_1 ... <var>_n) <term>)) ( $\lambda$  (<var>_0) (Curry ( $\lambda$  (<var>_1 ... <var>_n) <term>))))
  ((Curry <term>)) (Curry <term>))
  ((Curry (<term>_0 <term>_1)) ((Curry <term>_0) (Curry <term>_1)))
  ((Curry (<term>_0 <term>_1 <term>_2 ...)) (Curry ((<term>_0 <term>_1 <term>_2 ...))))
  ((Curry <var>) <var>))
```

The identifications  $\langle\text{term}\rangle \equiv \text{term}$  and  $(\lambda () \text{term}) \equiv \text{term}$  cannot be made in Scheme for it does not require that a function has an argument. For example  $((\lambda () 3)) \rightarrow 3$ , whereas  $(\lambda () 3) \rightarrow \#<\text{procedure}>$ . In the Lambda Calculus the identifications are quite natural, for  $\langle\text{term}\rangle$  and  $(\lambda () \text{term})$  are not allowed as curried lambda terms.

```
#lang scheme ; File test-curry.ss
(require redex "curry.ss" "uncurried-lc-grammar.ss" "curried-lc-grammar.ss")

(test-equal
  (term (Curry (( $\lambda$  (x x x) (x x x)) ( $\lambda$  (x x x) (x x x)) ( $\lambda$  (x x x) (x x x)))))
  (term ((( $\lambda$  (x) ( $\lambda$  (x) ( $\lambda$  (x) ((x x) x)))) ( $\lambda$  (x) ( $\lambda$  (x) ( $\lambda$  (x) ((x x) x)))) ( $\lambda$  (x) ( $\lambda$  (x) ( $\lambda$  (x) ((x x) x)))))))

(test-equal (term (Curry (((x)))) (term x))
  (test-equal (term (Curry ((( $\lambda$  () (((x) (y)))))))) (term (x y))))
```



```
(redex-check uncurried-lc-grammar <term> ; Check that Curry indeed forms curried terms.
  (term (Term? (Curry <term>))) #:attempts 2000 #:retries 100)

(test-results)
; Displays:
; redex-check: ...\\test-curry.ss:13 no counterexamples in 2000 attempts
; All 3 tests passed.
```

## Reduction

Up to now the grammar of the Lambda Calculus has been presented, together with some metafunctions and meta-predicates for curried terms. Nothing has been said about the semantics yet. In the Lambda Calculus evaluation is called ‘reduction’. A lambda term is reduced by contraction of redices until no more redices are left. There are two types of redices,  $\beta$ -redices and  $\eta$ -redices.

### $\beta$ -redices

A  $\beta$ -redex is a subterm of the form:

$((\lambda (\text{formal-argument}) \text{body}) \text{actual-argument})$ , where  $\text{body}$  and  $\text{actual-argument}$  are terms.

A  $\beta$ -redex is an application of an abstraction to an actual argument. Somehow the formal argument must be associated with the actual argument while the body of the abstraction is evaluated. In Scheme this is done by evaluation of the actual argument and binding the formal argument to the value of the actual argument. Subsequently the value of the body is computed in the environment of the abstraction to which the new binding is added. In the Lambda Calculus substitution is used. In principle this means that the above term can be contracted to the body, but with all free (id est non shadowed) occurrences of the formal argument replaced by the actual argument. For example:  $((\lambda (x) ((x\ x) (\lambda (x) x))) p) \rightarrow^{\beta} ((p\ p) (\lambda (x) x))$ , where the free occurrences of  $\mathbf{x}$  in body  $((\mathbf{x}\ \mathbf{x}) (\lambda (x) x))$ , id est the boldface ones, are replaced by actual argument  $p$ . However, there is a nasty pitfall:

$((\lambda (x) (\lambda (y) x)) y) \rightarrow^{\beta} (\lambda (y) y)$ , **wrong**.  
 $((\lambda (x) (\lambda (z) x)) y) \rightarrow^{\beta} (\lambda (z) y)$ , **ok**.

The problem is revealed by writing the free variable in boldface:

$((\lambda (x) (\lambda (y) x)) \mathbf{y}) \rightarrow^{\beta} (\lambda (y) \mathbf{y})$

A variable must not be substituted in a context where it would obtain another binding or would become bound whereas it was free. In this case correct substitution requires  $\alpha$ -conversion:

$((\lambda (x) (\lambda (y) x)) y) \rightarrow^{\alpha} ((\lambda (x) (\lambda (z) x)) y) \rightarrow^{\beta} (\lambda (z) y)$

Of course we use *redex* for the implementation of the  $\beta$ -reduction:

```
#lang scheme ; File beta-reductor.ss
(require redex "curried-lc-grammar.ss" "free-vars.ss")
(provide  $\beta$ -reductor Subst  $\beta$ nf?)

(define  $\beta$ -reductor
  (reduction-relation curried-lc-grammar
    (--> (in-hole <subterm> (( $\lambda$  (<var>) <term>_0) <term>_1))
      (in-hole <subterm> (Subst (<var> <term>_1) <term>_0) " $\beta$ "))))

(define-metafunction curried-lc-grammar  $\beta$ nf? : <term> -> <bool>
  (( $\beta$ nf? <term>) , (null? (apply-reduction-relation  $\beta$ -reductor (term <term>)))))
```

; Metafunction *Subst* does the substitution for  $\beta$ -contraction

```
(define-metafunction curried-lc-grammar Subst : (<var> <term>) <term> -> <term>
; Matching var: substitute <term>_0 for <var>_0.
((Subst (<var>_0 <term>_0) <var>_0) <term>_0)
; Non matching var: no substitution.
((Subst (<var>_0 <term>_0) <var>_1) <var>_1)
; Subst in abstr that shadows the var: no substitution in body.
((Subst (<var>_0 <term>_0) (λ (<var>_0) <term>_1)) (λ (<var>_0) <term>_1))
; Subst in abstr that does not shadow the var: β contraction (= substitution)
((Subst (<var>_0 <term>_0) (λ (<var>_1) <term>_1))
, (term-let ((Free (term (Free-vars <term>_0))))
  (if (member (term <var>_1) (term Free))
    (term-let ; α-conversion required.
      ((<new-var>
        (variable-not-in (term (<var>_0 Free (Free-vars <term>_1))) (term <var>_1)))
      (term
        (λ (<new-var>) ; <new-var> is the new name for <var>_1.
          (Subst (<var>_0 <term>_0) (Subst (<var>_1 <new-var>) <term>_1))))))
    (term (λ (<var>_1) (Subst (<var>_0 <term>_0) <term>_1)))))) ; α-conversion not required.
; Application: recur on operator and actual argument.
((Subst (<var>_0 <term>_0) (<term>_1 <term>_2))
  ((Subst (<var>_0 <term>_0) <term>_1) (Subst (<var>_0 <term>_0) <term>_2)))
```

```
#lang scheme ; file test-beta-reductor.ss
(require redex "beta-reductor.ss" "tracer.ss")

(parameterize ((reduction-steps-cutoff 100))
  (tracer β-reductor
    ((x y z) (x y z))
    ((λ (x) x) (λ (y) y))
    ((λ (x) (F x)) (λ (x) (F x)))
    (((λ (x) (y x)) x) (y x))
    (((λ (x y) x) y) (λ (p) y))
    (((λ (x) (λ (y) (x y))) z) (λ (y) (z y)))
    (((λ (x y z) (x y z)) a b c) (a b c))
    (((λ (x) (x x)) (λ (x) (x x))) #f)
    ((λ (p) ((λ (x) (x x)) (λ (x) (x x)))) #f)
    (((((λ (z) (z z)) (λ (x y) (x (x y)))) a) z) (a (a (a (a z)))))))

(parameterize ((reduction-steps-cutoff 10))
  (tracer β-reductor (((λ (x) (x x x)) (λ (x) (x x x))) #f)))
```

### η-redices

An η-redex is a lambda term of the form  $(\lambda (\langle \text{var} \rangle_0) (\langle \text{term} \rangle_0 \langle \text{var} \rangle_0))$ , where  $\langle \text{term} \rangle_0$  does not contain any free occurrence of the formal argument. With this condition, we have:

$$(\lambda (\langle \text{var} \rangle_0) (\langle \text{term} \rangle_0 \langle \text{var} \rangle_0)) \langle \text{term} \rangle_1 \rightarrow^\beta (\langle \text{term} \rangle_0 \langle \text{term} \rangle_1)$$

Therefore we can make the following contraction:

$$(\lambda (\langle \text{var} \rangle_0) (\langle \text{term} \rangle_0 \langle \text{var} \rangle_0)) \rightarrow^\eta \langle \text{term} \rangle_0, \text{ provided } \langle \text{var} \rangle_0 \text{ has no free occurrence in } \langle \text{term} \rangle_0.$$

This is an observation made by talking *about* the Lambda Calculus. It is not possible to derive this observation *within* the rules of *β-reductor* shown above. If we want to include η-contraction, the

above rule must be added to the semantics of the Lambda Calculus. We extend the  $\beta$ -reductor such as to include  $\eta$ -contraction

```
#lang scheme ; File beta-eta-reductor.ss
(require "curried-lc-grammar.ss" "beta-reductor.ss" "free-vars.ss" redex)
(provide  $\beta\eta$ -reductor  $\beta\eta$ nf?)

(define  $\beta\eta$ -reductor
  (extend-reduction-relation  $\beta$ -reductor curried-lc-grammar
    (--> (in-hole <subterm> ( $\lambda$  (<var>_0) (<term> <var>_0)))
      (in-hole <subterm> <term>)
      (side-condition (not (term (Var-free-in? <var>_0 <term>))))
      " $\eta$ ")))

(define-metafunction curried-lc-grammar  $\beta\eta$ nf? : <term> -> <bool>
  (( $\beta\eta$ nf? <term>) , (null? (apply-reduction-relation  $\beta\eta$ -reductor (term <term>)))))
```


```
#lang scheme ; file test-beta-eta-reductor.ss
(require redex "beta-eta-reductor.ss" "tracer.ss")

(parameterize ((reduction-steps-cutoff 100))
  (tracer  $\beta\eta$ -reductor
    ((x y z) (x y z))
    (( $\lambda$  (x) x) ( $\lambda$  (y) y))
    (( $\lambda$  (x) (x x)) ( $\lambda$  (y) (y y)))
    (( $\lambda$  (x) (F x)) F)
    (( $\lambda$  (x x) (F x)) ( $\lambda$  (x) F))
    (( $\lambda$  (x) (F x x)) ( $\lambda$  (x) (F x x)))
    ((( $\lambda$  (x) (y x)) x) (y x))
    ((( $\lambda$  (x y) x) y) ( $\lambda$  (p) y))
    ((( $\lambda$  (x) ( $\lambda$  (y) (x y))) z) z)
    ((( $\lambda$  (x y z) (x y z)) a b c) (a b c))
    ((( $\lambda$  (x) (x x)) ( $\lambda$  (x) (x x))) #f)
    (( $\lambda$  (p) (( $\lambda$  (x) (x x)) ( $\lambda$  (x) (x x)))) #f)
    ((((( $\lambda$  (z) (z z)) ( $\lambda$  (x y) (x (x y)))) a) z) (a (a (a (a z)))))))


(parameterize ((reduction-steps-cutoff 10))
  (tracer  $\beta\eta$ -reductor ((( $\lambda$  (x) (x x x)) ( $\lambda$  (x) (x x x))) #f)))
```

## Normal forms and reducibility


A normal form is a lambda term without redices, id est a term of which no subterm is a redex. Simpler said, a normal form is a lambda term that cannot be reduced any further.  $\beta$ -reduction consists of zero or more subsequent  $\beta$ -contractions, where needed with  $\alpha$ -conversion, until no more  $\beta$ -redices are left. The result is a  $\beta$ -normal form. It may still contain  $\eta$ -redices.  $\beta\eta$ -reduction consists of zero or more subsequent  $\beta$ - and  $\eta$ -contractions, where needed with  $\alpha$ -conversion. The result is a  $\beta\eta$ -normal form. A term is called reducible if it is a normal form or can be reduced to a normal form. If a lambda term has a normal form, the latter is uniquely defined modulo  $\alpha$ -congruence. This is a consequence of the Church-Rosser theorem, which holds for both  $\beta$ -reduction and  $\beta\eta$ -reduction. When running file [test-beta-eta-reductor.ss](#), you see that a lambda term can have zero one or more redices. The Church-Rosser theorem is: if a lambda term  $M$  can be reduced to a lambda term  $N_1$  and also to a lambda term  $N_2$ , then there is a lambda term  $L$  such that both  $N_1$  and  $N_2$  can be reduced to  $L$ . See figure 1. See [Barendregt](#) for a proof.

**Figure 1: The Church-Rosser theorem**

The two solid arrows imply the existence of  $L$  and the two dotted lines. Each arrow represents a sequence of zero or more contractions.

**Figure 2: Simple reduction graph**

Contraction of the underlined redex results in the very same term and makes the graph contain a cycle. Contraction of the only other redex immediately produces the normal form  $y$ .

**Figure 3:**

Infinite reduction graph for a term having a normal form. Hence the order of following the branches does matter.

## Reduction strategies

The Church-Rosser theorem implies that, in principle, the order of reduction is irrelevant. However, when contracting the underlined redex of the lambda term shown in figure 2, the original lambda term is obtained again, while the lambda term as a whole is a redex too and immediately reduces to the normal form  $y$ . In figure 3 a lambda term is shown with the same normal form, but with an infinite reduction graph for the underlined subterm. In both figure 2 and figure 3 a strategy is required that sooner or later takes the outer redex that branches down to  $y$ .

### Non deterministic or width first traversal

Procedure *traces* follows all possible reduction paths by traversing the graph in width first order. See file [width-first-traversal.ss](#) for a simple example of this method. This strategy ensures that given a reducible lambda term, the normal form will be found. It is as though *traces*, when inspecting a lambda term, contracts all its redices in parallel and follows all branches to the reducts simultaneously.

### One step or deterministic reduction strategies

A one step strategy is one that chooses one branch only. The result is a linear path without bifurcations. The branch is determined by a function, say  $F$ , such that given a non normal form  $\langle \text{term} \rangle$ ,  $F(\langle \text{term} \rangle)$  can be obtained from  $\langle \text{term} \rangle$  by one single contraction. Examples are: normal order, applicative order and random order. Given an initial term, the followed path is:  $\langle \text{term} \rangle \rightarrow F(\langle \text{term} \rangle) \rightarrow F(F(\langle \text{term} \rangle)) \equiv F^2(\langle \text{term} \rangle) \rightarrow F(F^2(\langle \text{term} \rangle)) \equiv F^3(\langle \text{term} \rangle) \rightarrow \dots \rightarrow F^n(\langle \text{term} \rangle) \dots$ , until a normal form is

obtained (or non ending if the reduction strategy fails or the term is irreducible)

### Normal order reduction strategy

For the normal order strategy function  $F$  selects the leftmost redex, id est the redex whose left parenthesis is left of those of all other redices. When given a reducible term, the normal order strategy does lead to the normal form within a finite number of reduction steps. Let  $N$  be the normal form of a reducible term  $M$ . If this strategy would not find the normal form, then every reduction from  $M$  to  $N$  would require one or more steps involving the contraction of a non leftmost redex. Let  $M_i \rightarrow M_{i+1}$  be such a non leftmost reduction step. Then  $M_{i+1}$  still has the leftmost redex of  $M_i$ , possibly internally contracted. More contractions may follow, possibly affecting the subterms of the still not contracted leftmost redex. A non leftmost contraction cannot contract the still uncontracted (possibly internally modified) leftmost redex. Eventually we must contract this leftmost redex in order to obtain the normal form. We may as well do so for the step  $M_i \rightarrow M_{i+1}$ .

### Applicative order reduction strategy

The normal order reduction strategy is a safe one, but not always efficient. For example:

$$\begin{aligned} & ((\lambda (x) (x x)) ((\lambda (x) x) (\lambda (x) x))) \rightarrow \\ & (((\lambda (x) x) (\lambda (x) x)) ((\lambda (x) x) (\lambda (x) x))) \rightarrow \\ & ((\lambda (x) x) ((\lambda (x) x) (\lambda (x) x))) \rightarrow \\ & ((\lambda (x) x) (\lambda (x) x)) \rightarrow (\lambda (x) x) \end{aligned}$$

In each line the redex being contracted is underlined. We see that subterm  $((\lambda (x) x) (\lambda (x) x))$  is reduced twice. This is avoided by the applicative order strategy. For this strategy we shall use  $\beta$ -reduction only (without  $\eta$ -contractions). We still consider the leftmost redex, but before contracting it, the formal argument is reduced. This gives:

$$((\lambda (x) (x x)) ((\lambda (x) x) (\lambda (x) x))) \rightarrow ((\lambda (x) (x x)) (\lambda (x) x)) \rightarrow ((\lambda (x) x) (\lambda (x) x)) \rightarrow (\lambda (x) x)$$

This is only one step less than for normal order reduction, but in more complicated examples the gain of efficiency can be huge. A drawback of applicative order reduction is that it does not always find the normal form. Figures 2 and 3 show examples of reducible terms that cannot be reduced to their normal forms in applicative order.

### Random order reduction strategy

The next redex to be contracted can be chosen randomly from all redices. With a good random selector this strategy almost surely leads to the normal form, provided the initial term is reducible. The reason is that with a good random selector, there is a fair chance that after ignoring a leftmost redex for a while, it will eventually be chosen.

## Normal order $\beta$ -reduction

In this section two programs are shown for normal order  $\beta$ -reduction. The first one uses a metafunction for the reduction proper. The second one uses a language and reduction-relation that recognize leftmost  $\beta$ -redices.

```
#lang scheme ; File normal-order-beta-reductor-version-1.ss

(require redex "curried-lc-grammar.ss")
(require (only-in "beta-reductor.ss" Subst))
(provide normal-order-β-reductor-version-1 βNf? β-redex?)

(define normal-order-β-reductor-version-1
  (reduction-relation curried-lc-grammar
    (--> <term> (Contract-once <term>)
      (side-condition (not (term (βNf? <term>)))))))
```

```

(define-metafunction curried-lc-grammar  $\beta$ Nf? : <term> -> <bool>
  (( $\beta$ Nf? <var>) #t)
  (( $\beta$ Nf? ( $\lambda$  (<var>) <term>)) ( $\beta$ Nf? <term>))
  (( $\beta$ Nf? (<term>_0 <term>_1))
   , (and (not (term (Abstr? <term>_0)))
          (term ( $\beta$ Nf? <term>_0))
          (term ( $\beta$ Nf? <term>_1)))))

(define-metafunction curried-lc-grammar  $\beta$ -redex? : <term> -> <bool>
  (( $\beta$ -redex? (( $\lambda$  (<var>) <term>_0) <term>_1)) #t)
  (( $\beta$ -redex? any) #f))

(define-metafunction curried-lc-grammar Contract-once : <term> -> <term>
  ((Contract-once (side-condition <term> (term ( $\beta$ -redex? <term>))))
   (Contract- $\beta$ -redex <term>))
  ((Contract-once ( $\lambda$  (<var>) <term>))
   ( $\lambda$  (<var>) (Contract-once <term>)))
  ((Contract-once ((side-condition <term>_0 (not (term ( $\beta$ Nf? <term>_0)))) <term>_1))
   ((Contract-once <term>_0) <term>_1))
  ((Contract-once (<term>_0 <term>_1))
   (<term>_0 (Contract-once <term>_1)))

(define-metafunction curried-lc-grammar Contract- $\beta$ -redex : <term> -> <term>
  ((Contract- $\beta$ -redex (( $\lambda$  (<var>_0) <term>_0) <term>_1))
   (Subst (<var>_0 <term>_1) <term>_0)))

```

```

#lang scheme ; File test-normal-order-beta-reductor-version-1.ss
(require (only-in "curried-lc-grammar.ss" curried-lc-grammar))
(require "normal-order-beta-reductor-version-1.ss")
(require "normal-order-test.ss")
(normal-order-test curried-lc-grammar normal-order- $\beta$ -reductor-version-1  $\beta$ Nf?)

```

```

#lang scheme ; File normal-order-test.ss
(require redex "tracer.ss")
(provide normal-order-test)

(define-syntax normal-order-test
  (syntax-rules ()
    ((_ grammar reductor  $\beta$ Nf?)
     (begin
      (tracer reductor
        (((( $\lambda$  (x y) x) y) z) (((( $\lambda$  (z) (z z)) ( $\lambda$  (x y) (x (x y)))) a) z)) (y (a (a (a (a z))))))
        (((( $\lambda$  (z) (z z)) ( $\lambda$  (x y) (x (x y)))) a) z) (a (a (a (a z))))))
        (( $\lambda$  (x) (y x)) (y x)) (( $\lambda$  (x) (y x)) (y x))
        ((( $\lambda$  (x) x) y) y)
        ((( $\lambda$  (x y z) (x y z)) a b c) (a b c))
        ((( $\lambda$  (z) ((a b) z)) c) (a b c))
        ((( $\lambda$  (x) (x x)) ( $\lambda$  (x) (x x))) #f)))

```

```

(redex-check grammar <term>
  (let*
    ((nf? (term (βNf? <term>)))
     (redices (apply-reduction-relation reductor (term <term>))))
    (cond
      ((and (pair? redices) (pair? (cdr redices)))
       (error 'test "more than one left most redex for term ~s"
        (term <term>)))
      ((and nf? (pair? redices))
       (error 'test "(βNf? ~s)->#t but reductor found redex ~s."
        (term <term>) (car redices)))
      ((and (not nf?) (null? redices))
       (error 'test "(βNf? ~s)->#f but reductor found no redex."
        (term <term>)))))) #:attempts 10000 #:retries 100))))

```

```

#lang scheme ; File normal-order-beta-reductor-version-2.ss
(require redex)
(require (only-in "curried-lc-grammar.ss" curried-lc-grammar))
(require (only-in "free-vars.ss" Free-vars))
(require (only-in "beta-reductor.ss" Subst))
(provide normal-order-β-reductor-version-2 βNf? curried-lc-language)

(define-extended-language curried-lc-language curried-lc-grammar
  (<nf> (λ (<var>) <nf>) <non-abstr-nf>)
  (<non-abstr-nf> <var> <non-abstr-nf> <nf>))
  (<hole> hole (λ (<var>) <hole>) <non-abstr-nf> <hole>)
  (side-condition (<hole>_1 <term>) (not (term (Abstr? <hole>_1))))))

(define normal-order-β-reductor-version-2
  (reduction-relation curried-lc-language
    (--> (in-hole <hole>) ((λ (<var>) <term>_1) <term>_2))
      (in-hole <hole>) (Subst (<var> <term>_2) <term>_1))))

(define-metafunction curried-lc-language Abstr? : any -> any
  ((Abstr? (λ (<var>) any)) #t) ((Abstr? any) #f))

(define-metafunction curried-lc-language βNf? : any -> any
  ((βNf? <nf>) #t) ((βNf? any) #f))

```

```

#lang scheme ; File test-normal-order-beta-reductor-version-2.ss
(require "normal-order-beta-reductor-version-2.ss")
(require "normal-order-test.ss")
(normal-order-test curried-lc-language normal-order-β-reductor-version-2 βNf?)

```

## Equality relation on lambda terms

The equality relation on lambda terms is the compatible, reflexive, symmetric and transitive closure of  $\alpha$ -congruence and the  $\beta$ - or  $\beta\eta$ -reduction-relation. This means the following list of axioms:



0	$\langle \text{term} \rangle_0 \rightarrow^\alpha \langle \text{term} \rangle_1 \Rightarrow \langle \text{term} \rangle_0 = \langle \text{term} \rangle_1$	$\alpha$ -congruence.
1	$\langle \text{term} \rangle_0 \rightarrow^\beta \langle \text{term} \rangle_1 \Rightarrow \langle \text{term} \rangle_0 = \langle \text{term} \rangle_1$	Single step $\beta$ -contraction.
2	$\langle \text{term} \rangle_0 = \langle \text{term} \rangle_0$	Reflexive closure.
3	$\langle \text{term} \rangle_0 = \langle \text{term} \rangle_1 \Rightarrow \langle \text{term} \rangle_1 = \langle \text{term} \rangle_0$	Symmetric closure.
4	$\langle \text{term} \rangle_0 = \langle \text{term} \rangle_1 \wedge \langle \text{term} \rangle_1 = \langle \text{term} \rangle_2 \Rightarrow \langle \text{term} \rangle_0 = \langle \text{term} \rangle_2$	Transitive closure.
5	$\langle \text{term} \rangle_0 = \langle \text{term} \rangle_1 \Rightarrow (\lambda (\langle \text{var} \rangle_0) \langle \text{term} \rangle_0) = (\lambda (\langle \text{var} \rangle_0) \langle \text{term} \rangle_1)$	Compatibility with abstraction.
6	$\langle \text{term} \rangle_0 = \langle \text{term} \rangle_1 \Rightarrow (\langle \text{term} \rangle_0 \langle \text{term} \rangle_2) = (\langle \text{term} \rangle_1 \langle \text{term} \rangle_2)$	Compatibility with application.
7	$\langle \text{term} \rangle_0 = \langle \text{term} \rangle_1 \Rightarrow (\langle \text{term} \rangle_2 \langle \text{term} \rangle_0) = (\langle \text{term} \rangle_2 \langle \text{term} \rangle_1)$	Compatibility with application.

Axiom 5 is also called ‘rule  $\xi$ ’. We may also add:

8	$\langle \text{term} \rangle_0 \langle \text{var} \rangle_0 = \langle \text{term} \rangle_1 \langle \text{var} \rangle_0 \Rightarrow \langle \text{term} \rangle_0 = \langle \text{term} \rangle_1$ provided $\langle \text{var} \rangle_0$ has no free occurrences in $\langle \text{term} \rangle_0$ nor in $\langle \text{term} \rangle_1$ .	Extensionality.
---	--	-----------------

Addition of extensionality (axiom 8) is equivalent with adding  $\eta$ -conversion.

Proof: extensionality  $\Rightarrow \eta$ -conversion (in *about* mode)

If  $\langle \text{var} \rangle_0$  has no free occurrences in  $\langle \text{term} \rangle_0$ , then we have  $((\lambda (\langle \text{var} \rangle_0) (\langle \text{term} \rangle_0 \langle \text{var} \rangle_0)) \langle \text{var} \rangle_0) = (\langle \text{term} \rangle_0 \langle \text{var} \rangle_0)$ . Now extensionality gives  $(\lambda (\langle \text{var} \rangle_0) (\langle \text{term} \rangle_0 \langle \text{var} \rangle_0)) = \langle \text{term} \rangle_0$ . QED.

Proof:  $\eta$ -conversion  $\Rightarrow$  extensionality (in *about* mode)

Suppose  $\langle \text{term} \rangle_0 \langle \text{var} \rangle_0 = \langle \text{term} \rangle_1 \langle \text{var} \rangle_0$ , where  $\langle \text{var} \rangle_0$  has no free occurrences in  $\langle \text{term} \rangle_0$  nor in  $\langle \text{term} \rangle_1$ . Then by rule  $\xi$ :  $(\lambda (\langle \text{var} \rangle_0) (\langle \text{term} \rangle_0 \langle \text{var} \rangle_0)) = (\lambda (\langle \text{var} \rangle_0) (\langle \text{term} \rangle_1 \langle \text{var} \rangle_0))$ .  $\eta$ -contraction of the left and right hand sides gives:  $\langle \text{term} \rangle_0 = \langle \text{term} \rangle_1$ . QED.

With addition of the extensionality, the Lambda Calculus has the following property of completeness: let  $M$  and  $N$  be two reducible lambda terms and  $M \neq N$ ; then addition of  $M=N$  to axioms 1-8 produces a degenerated calculus in which all lambda terms are equal to each other. It is like adding the axiom  $0=1$  to number theory, making all numbers equal to each other. See [Barendregt](#) for proof.

## Applicative order Y combinator in Scheme

Y combinators are used for the creation of recursive function without the use of forms like *define*, *letrec*, named *let* or other forms that bind variables before their values have been established. Before presenting a more formal introduction to Y combinators, this section shows a derivation in pure Scheme. Although this derivation requires more text than the more formal introduction in the next section, it may nevertheless be a good preparation. <sup>(6)</sup>

#lang scheme	; <a href="#">File y-derivation-in-scheme.ss</a>
Preparation of a recursive function without <i>define</i> , named <i>let</i> , <i>letrec</i> or any form alike. First look at a simple way to define function factorial using <i>define</i> :	
; Version 1	
(define factorial-1 (lambda (n) (if (zero? n) 1 (* n (factorial-1 (sub1 n))))))	
(factorial-1 4)	
Add a layer of abstraction (why do we want an extra layer of abstraction? Read on and see)	
; Version 2	
(define make-factorial-2	
(lambda ()	
(lambda (n) (if (zero? n) 1 (* n ((make-factorial-2) (sub1 n))))))	
(define factorial-2 (make-factorial-2))	
(factorial-2 4)	

6 With thanks to Daniel P. Friedman and Matthias Felleisen permitting me to use an adapted version of their approach in ‘The Little LISPer’ c.q. ‘[The Little Schemer](#)’.

Function *factorial-2* has no argument, but we can give it a dummy argument. Why this is done will become clear within three steps.

### ; Version 3

```
(define make-factorial-3
  (λ (dummy-arg)
    (λ (n) (if (zero? n) 1 (* n ((make-factorial-3 dummy-arg) (sub1 n))))))

(define factorial-3 (make-factorial-3 "dummy"))
(factorial-3 4)
```

In version 3 the dummy argument is never used. It is just passed around. It does not matter what actual argument we provide for the dummy formal argument. We may as well pass *make-factorial-3* to itself. Why we want to do that? This will become clear within two more steps. The dummy argument can be renamed such as to indicate that its value will be the factorial-maker itself.

### ; Version 4

```
(define make-factorial-4
  (λ (make-factorial)
    (λ (n) (if (zero? n) 1 (* n ((make-factorial-4 make-factorial) (sub1 n))))))

(define factorial-4 (make-factorial-4 make-factorial-4))
(factorial-4 4)
```

In *make-factorial-4* we have subexpression `(make-factorial-4 make-factorial)` but *make-factorial-4* is called with itself for its actual argument `make-factorial`. The two variables have the same value. Hence we can replace the subexpression by `(make-factorial make-factorial)`.

### ; Version 5

```
(define make-factorial-5
  (λ (make-factorial)
    (λ (n) (if (zero? n) 1 (* n ((make-factorial make-factorial) (sub1 n))))))

(define factorial-5 (make-factorial-5 make-factorial-5))
(factorial-5 4)
```

Now we see why it made sense to apply *make-factorial* to itself. *Make-factorial-5* no longer is recursive, or is it? Anyway, function `(λ (make-factorial) ...)` no longer has a recursive reference to variable `make-factorial-5`.

Function `(λ (n) ...)` contains the subexpression `(make-factorial make-factorial)`, where we would like to write `factorial`. This can be done by binding the value of the subexpression to a variable of this name.

### ; Version 6

```
; (define make-factorial-6
;   (λ (make-factorial)
;     ((λ (factorial)
;        (λ (n) (if (zero? n) 1 (* n (factorial (sub1 n))))))
;      (make-factorial make-factorial))))

; (define factorial-6 (make-factorial-6 make-factorial-6))
; (factorial-6 4)
```

Version 6 has been commented out for good reason. Copy the example into a fresh definitions window, uncomment it and use Debug to see what happens. Function *make-factorial-6* immediately applies, *make-factorial* to itself. But we have given *make-factorial-6* as the value for `make-factorial`. Hence *make-factorial-6* is applied to itself, but this application makes *make-factorial-6* apply itself to itself immediately again and again and again ... What has happened during the trans-

formation from version 5 to version 6? In version 5 the self application (`make-factorial make-factorial`) is in the `else`-part of an `if`-form. It is evaluated only when needed. At the bottom of the recursion, it is not evaluated. However, in version 6 the self-application is evaluated unconditionally for it has been lifted out of the `if`-form. We have to postpone the self application until it is needed in the `else`-part of the `if`-form. This can be done by wrapping it in a lambda-form. If we have an expression  $F$  that produces a function of one argument, the function  $(\lambda (x) (F x))$  exactly does the same, provided evaluation of  $F$  produces no side effects and  $F$  does not refer to variable  $x$ . If  $F$  does depend on a variable called ' $x$ ', we can take  $(\lambda (y) (F y))$  or choose any other formal argument that does not conflict with the variables used in  $F$ .  $F$  is not evaluated before  $(\lambda (x) (F x))$  is called. That is precisely what a function is for. It is a recipe. We don't have to start cooking right away. We can wait until we are hungry. The appetite is in the `else`-part if the `if`-form.

### ; Version 7

```
(define make-factorial-7
  (lambda (make-factorial)
    ((lambda (factorial) (lambda (n) (if (zero? n) 1 (* n (factorial (sub1 n))))))
     (lambda (n) ((make-factorial make-factorial) n)))))

(define factorial-7 (make-factorial-7 make-factorial-7))
(factorial-7 4)
```

Now we can put the definition of *make-factorial-7* into the self application that defines *factorial-7*:

### ; Version 8

```
(define factorial-8
  ((lambda (f) (f f)) ; does the self application.
   (lambda (make-factorial)
     ((lambda (factorial) (lambda (n) (if (zero? n) 1 (* n (factorial (sub1 n))))))
      (lambda (n) ((make-factorial make-factorial) n)))))

(factorial-8 4)
```

Function  $(\lambda (factorial) \dots)$  does not refer to `make-factorial`. Therefore it can be lifted to out. We call it  $m$ .

### ; Version 9

```
(define factorial-9
  ((lambda (m)
    ((lambda (f) (f f))
     (lambda (make-factorial) (m (lambda (n) ((make-factorial make-factorial) n)))))
   (lambda (factorial) (lambda (n) (if (zero? n) 1 (* n (factorial (sub1 n)))))))

(factorial-9 4)
```

Now function  $(\lambda (m) \dots)$  is independent from the algorithm used for the factorial function. It can be used for other functions too. Therefore it is better to change the name `make-factorial` into a more general one. We choose  $g$ .

### ; Version 10

```
(define factorial-10
  ((lambda (m) ((lambda (f) (f f)) (lambda (g) (m (lambda (n) ((g g) n))))))
   (lambda (factorial) (lambda (n) (if (zero? n) 1 (* n (factorial (sub1 n)))))))

(factorial-10 4)

(define length-10
  ((lambda (m) ((lambda (f) (f f)) (lambda (g) (m (lambda (n) ((g g) n))))))
   (lambda (length) (lambda (lyst) (if (null? lyst) 0 (add1 (length (cdr lyst)))))))
```





DrScheme's definition window and use Debug to see what happens. The reason that the above factorial function does not work is that Scheme evaluates in applicative order.  $\mathbf{Y}$  is not suited for this strategy of reduction. This has already been discussed in the previous section. For applicative order reduction use  $(\lambda (m) ((\lambda (f) (m (f f))) (\lambda (f) (m (\underline{\lambda (x) ((f f) x))}))))$ .

## Undecidability

Write  $\mathbf{T} \equiv (\lambda (x y) x)$  and  $\mathbf{F} \equiv (\lambda (x y) y)$ . A lambda term, say  $P$ , is a predicate if for every arbitrary lambda term  $M$ , either  $(P M) = \mathbf{T}$  or  $(P M) = \mathbf{F}$ , or  $(P M)$  has no normal form.  $P$  is complete if  $(P M)$  has a normal form for every arbitrary  $M$ , meaning that  $(P M)$  is either  $\mathbf{T}$  or  $\mathbf{F}$ .  $P$  is not trivial if there are two terms  $T$  and  $F$  such that  $(P T) = \mathbf{T}$  and  $(P F) = \mathbf{F}$ .

### Theorem

There is no non trivial complete predicate.

### Proof

The theorem is proven by reductio ad absurdum. Suppose that  $P$  is a non trivial complete predicate. Then there is a  $T$  and an  $F$  such that  $(P T) = \mathbf{T}$  and  $(P F) = \mathbf{F}$ . Now define:

$$\begin{aligned} N &\equiv (\lambda (x) (P \times F T)) & N \text{ for 'negation'} \\ A &\equiv (\mathbf{Y} N) = (N (\mathbf{Y} N)) = (N A) & A \text{ for 'absurdum'} \end{aligned}$$

Because  $P$  is supposed to be complete we have either:  $(P A) = \mathbf{T}$  or  $(P A) = \mathbf{F}$ . However:

$(P A) = \mathbf{T} \Rightarrow \mathbf{T} = (P A) = (P (N A)) = (P (P A F T)) = (P (\mathbf{T} F T)) = (P F) = \mathbf{F}$ , a contradiction.

$(P A) = \mathbf{F} \Rightarrow \mathbf{F} = (P A) = (P (N A)) = (P (P A F T)) = (P (\mathbf{F} F T)) = (P T) = \mathbf{T}$ , a contradiction.

Hence  $P$  does not exist. QED.

### Corollary

From the above theorem it immediately follows that there is no complete non trivial predicate that can decide whether or not its argument has a normal form ([halting problem](#)). It means that there is no program that for all programs can decide whether or not it will produce a result within a finite number of steps.

### Corollary

There is no term  $E$  such that  $(E M N) = \mathbf{T}$  if  $N = M$  and  $(E M N) = \mathbf{F}$  if  $N \neq M$ , for this would mean that for every  $M$  the term  $(E M)$  would have to be a complete non trivial predicate. If we restrict the domain to reducible terms  $M$  and  $N$ , then it is possible to construct  $E$  such that  $(E M N) = \mathbf{T}$  if  $N = M$  and  $(E M N) = \mathbf{F}$  if  $N \neq M$ , for example by reducing  $M$  and  $N$  and check whether or not they are  $\alpha$ -congruent. However, the set of reducible terms is not well defined for we have no predicate for that property, nor within the Lambda Calculus, nor when reasoning about the Lambda Calculus.

One can even prove something stronger than the above theorem. The range of a lambda term  $M$  is the set  $\{(M N), N \text{ a lambda term}\}$ . A range either consists of one single element or is infinite. See [Barendregt](#) for a proof.

## Applicative order reduction

Normal order reduction is effective, but not efficient. For example:  $((\lambda (x) (x x)) \langle \text{argument} \rangle) \rightarrow (\langle \text{argument} \rangle \langle \text{argument} \rangle)$ . This means that both occurrences of the  $\langle \text{argument} \rangle$  must yet be reduced. This can be avoided by reducing the  $\langle \text{argument} \rangle$  before contracting the outermost redex. However, this strategy does not always find the normal form of reducible terms. For example:

$((\lambda (x) y) (\underline{(\lambda (x) (x x)) (\lambda (x) (x x))})) \xrightarrow{\text{applicative-order}} ((\lambda (x) y) ((\lambda (x) (x x)) (\lambda (x) (x x))))$  because

$\underline{((\lambda (x) (x x)) (\lambda (x) (x x)))} \rightarrow \underline{((\lambda (x) (x x)) (\lambda (x) (x x)))}$ . Yet the term is reducible. Its normal form is  $y$ .

See figure 2. This means that the applicative order strategy does not find the normal forms of all reducible terms. We have to limit the terms to be reduced to those in which a function is never applied to an irreducible argument. Another problem is the if-form. It is essential that in  $(\text{if } \langle \text{test} \rangle \langle \text{then-part} \rangle \langle \text{else-part} \rangle)$  the two alternatives are not evaluated before the test has decided

which one to choose. Therefore *if* is necessarily a syntactic form in Scheme. Nevertheless we can write the *if*-form in terms of functions only. We only have to postpone the evaluation of the two alternatives. This can be done by wrapping them in abstractions:

```
((test) (λ (ignore) <then-part>) (λ (ignore) <else-part>)) <to-be-ignored>
```

The test, which must yield either **T** or **F**, selects one of the two abstractions. The selected one is applied to <to-be-ignored>.

## Lazy-evaluator

Lazy reduction is yet another deterministic reduction strategy. When a function is applied to an actual argument, as in  $(\lambda (\langle \text{formal-argument} \rangle) \langle \text{body} \rangle) \langle \text{actual-argument} \rangle$ , we can attach a label to each free occurrence of the formal argument within the body. The label refers to the still uncontracted actual argument. For each application a fresh label must be chosen. Now the body can be reduced. If during the reduction of the body the formal argument is referenced, the corresponding actual argument is reduced and its normal form substituted for every variable that has the same label. This has the advantage of postponing the reduction of actual arguments that will never be needed, but avoids multiple reduction of the same argument. For reasons of efficiency a lazy evaluator will be used in the remainder of this essay. It is like Scheme, but with the following modifications:

- 0 Each uncurried term is evaluated as though fully curried. This allows uncurried shorthand.
- 1 Free variables are allowed and are self-evaluating.
- 2 Abstractions are evaluated to Scheme procedures.
- 3 Evaluation of an application does not involve the evaluation of the actual argument. The actual argument is wrapped in a promise and the latter is passed to the procedure. The Scheme procedures mentioned in item 2 know that their argument may be a promise rather than the value itself.

Syntax *ev* in file *lazy-reductor.ss* shows the implementation. It is not a reductor in the proper sense, because it may return a Scheme procedure, which is not a lambda term.

Syntax *def* allows the preparation of environments to be used during subsequent calls to syntax *ev*.  $(\text{def } \langle \text{var} \rangle \langle \text{term} \rangle)$  differs from  $(\text{define } \langle \text{var} \rangle \langle \text{term} \rangle)$  because *def* promises to evaluate the term in the an environment that does not include the new binding. It does not allow forward or recursive references to any variable. *Def* cannot construct self reference or recursion as can be done with *define* or *letrec*. *Def* may be used to redefine a variable. The new definition shadows the old one, but does not affect any definition that was made before the variable was shadowed. A sequence of *def*-forms preceding an *ev*-form has the same effect as  $(\text{ev } (\text{let}^* ((\langle \text{var} \rangle \langle \text{term} \rangle) \dots) \langle \text{term} \rangle))$ , where the *let\**-form is expanded in terms of lambda, which can be done as follows:

```
#lang scheme ; File let-star-in-terms-of-lambda.ss

(define-syntax let*
  (syntax-rules ()
    ((let* () . body) ((λ () . body)))
    ((let* ((var expr) binding ...) . body)
     ((λ (var) (let* (binding ...) . body)) expr))))
```

```
#lang scheme ; File lazy-evaluator.ss
```

Syntax:  $(\text{ev } \langle \text{term} \rangle) \rightarrow \text{value of the term}$

Curries the term and evaluates it in the current environment. Free variables are allowed and self-evaluating. When an application is evaluated, the operator must be a function. The actual-argument is not evaluated, but wrapped in a promise. The value of an abstraction is a procedure of one argument. If this procedure receives a promise, the promise will be forced when needed.



**Syntax:** (def <var> <term>)

First promises to evaluate the term in the current environment, then forms a new current environment by adding the binding to the old current environment. It is allowed to shadow an existing binding. Initially the current environment is empty.

**Procedure:** (list-vars) → list of variables

Returns the list of all variables bound in the current environment. Shadowed variables are included.

(clear-env) → void

Clears the current environment.

```
(define-syntax (def stx)
  (syntax-case stx ()
    ((_ var x) #'(extend-env 'var (let ((e env)) (lazy (ev-aux 'x e))))))

(define-syntax ev (syntax-rules () ((_ x) (ev-aux 'x env))))

(define (ev-aux x env)
  (cond
    ((var? x) (force (lookup x env)))
    ((abstr? x) (make-fun (cadr x) (caddr x) env))
    ((appl? x) (make-appl (ev-aux (car x) env) (cdr x) env))
    (else (error 'ev "incorrect (sub)term ~s" x))))

(define (make-fun vars body env)
  (if (null? vars) (ev-aux body env)
      (let ((var (car vars)) (vars (cdr vars)))
        (λ (x) (make-fun vars body (cons (cons var x) env))))))

(define (make-appl op args env)
  (if (null? args) op
      (let ((arg (car args)) (args (cdr args)))
        (make-appl (op (lazy (ev-aux arg env))) args env))))

(define (abstr? x) (and (list? x) (= (length x) 3) (eq? (car x) 'λ) (formals? (cadr x))))
(define (formals? x) (and (list? x) (andmap var? x)))
(define (appl? x) (and (list? x) (pair? x) (not (eq? (car x) 'λ))))
(define (var? x) (and (symbol? x) (not (eq? x 'λ))))
(define env '())
(define (clear-env) (set! env '()))
(define (list-vars) (map car env))
(define (list-env) env)
(define (lookup var env) (let ((binding (assq var env))) (if binding (cdr binding) var)))
(define (extend-env var value) (set! env (cons (cons var value) env)))
(provide ev def list-env list-vars clear-env)
```

## Booleans

The main use of Booleans in a programming language is for Boolean logic and for the test in an if-form in order to choose one out of two options. Therefore we represent true and false by functions of two arguments returning the first or the second argument. Every function with this property is a Boolean. Other values cannot be used as Booleans. In fact the functions will be curried, of course. In our terminology a function of two arguments is a function of one argument producing another function of one argument.

```
#lang scheme
(require "lazy-evaluator.ss")
```

; [File booleans.ss](#)

```
(def True  (λ (x y) x)) ; ≡ (λ (x) (λ (y) x))
(def False (λ (x y) y)) ; ≡ (λ (x) (λ (y) y))
(def If    (λ (x y z) (x y z))) ; A function! Unnecessary, but it may improve readability.
(def And   (λ (x y) (If x y False)))
(def Or    (λ (x y) (If x x y)))
(def Not   (λ (x) (If x False True)))
```

```
#lang scheme ; File test-booleans.ss
(require redex "booleans.ss" "curried-scheme.ss")
(test-equal (curried-scheme (True yes no)) 'yes)
(test-equal (curried-scheme (False yes no)) 'no)
(test-equal (curried-scheme (Not False yes no)) 'yes)
(test-equal (curried-scheme (Not True yes no)) 'no)
(test-equal (curried-scheme (And True True yes no)) 'yes)
(test-equal (curried-scheme (And True False yes no)) 'no)
(test-equal (curried-scheme (And False True yes no)) 'no)
(test-equal (curried-scheme (And False False yes no)) 'no)
(test-equal (curried-scheme (Or True True yes no)) 'yes)
(test-equal (curried-scheme (Or True False yes no)) 'yes)
(test-equal (curried-scheme (Or False True yes no)) 'yes)
(test-equal (curried-scheme (Or False False yes no)) 'no)
(test-equal (curried-scheme (If True yes no)) 'yes)
(test-equal (curried-scheme (If False yes no)) 'no)
(test-equal (curried-scheme (If (Not False) yes no)) 'yes)
(test-equal (curried-scheme (If (Not True) yes no)) 'no)
(test-equal (curried-scheme (If (And True True) yes no)) 'yes)
(test-equal (curried-scheme (If (And True False) yes no)) 'no)
(test-equal (curried-scheme (If (And False True) yes no)) 'no)
(test-equal (curried-scheme (If (And False False) yes no)) 'no)
(test-equal (curried-scheme (If (Or True True) yes no)) 'yes)
(test-equal (curried-scheme (If (Or True False) yes no)) 'yes)
(test-equal (curried-scheme (If (Or False True) yes no)) 'yes)
(test-equal (curried-scheme (If (Or False False) yes no)) 'no)
(test-results) ; Display: All 24 tests passed.
```

## Pairs

How shall we represent pairs? By functions of course, for we have no other equipment available. The essential property of a pair is that it must allow recollection of its car and its cdr. Therefore a pair will be a function that expects a Boolean for its argument and returns the car or cdr depending on the Boolean.

```
#lang scheme ; File pairs.ss
(require "booleans.ss" "lazy-evaluator.ss")
(def Cons (λ (x y z) (If z x y))) ; ≡ (λ (x y) (λ (z) (If z x y)))
(def Car  (λ (x) (x True)))
(def Cdr  (λ (x) (x False)))
```

Now indeed:

```
(Car (Cons <x> <y>)) = (Car (λ (z) (z <x> <y>))) = ((λ (z) (z <x> <y>)) True) = (True <x> <y>) = <x>
(Cdr (Cons <x> <y>)) = (Cdr (λ (z) (z <x> <y>))) = ((λ (z) (z <x> <y>)) False) = (False <x> <y>) = <y>
```

```
#lang scheme ; File test-pairs.ss
(require redex "pairs.ss" "lazy-evaluator.ss")
(test-equal (ev (Car (Cons yes no))) 'yes)
(test-equal (ev (Cdr (Cons yes no))) 'no)
(test-results)
```

## Numbers

There are several ways to represent natural numbers in Lambda Calculus. The representation of a number is called a numeral. A numeral corresponding to natural number  $n$  will be written as  $[n]$ .

### Numbers represented by pairs

$[0] \equiv (\lambda (x) x)$ ,  $[n+1] \equiv (\text{Cons } \text{False } [n])$

```
#lang scheme ; File pair-numerals.ss
(require "pairs.ss" "booleans.ss" "lazy-evaluator.ss")

(def Zero (λ (x) x))
(def Zero? Car)
(def Add1 (Cons False))
(def Sub1 Cdr)
```

Now we must show that *Zero*, *Zero?*, *Add1* and *Sub1* form an adequate number system. We do this by proving that the [axioms of Peano](#) are satisfied. Here we are reasoning *about* the system!

0 Consider *Zero* to be a number. For every number  $\langle n \rangle$ , consider  $(\text{Add1 } \langle n \rangle)$  to be a number too. Also assume that numbers are not made in any other way. This ensures that the axiom of mathematical induction applies.

1  $(\text{Zero? Zero}) = (\text{Car Zero}) = (\text{Zero True}) = ((\lambda (x) x) \text{True}) = \text{True}$  and  
 $(\text{Zero? (Add1 } \langle n \rangle)) = (\text{Car (Cons False } \langle n \rangle)) = \text{False}$ .

Hence:  $(\text{Add1 } \langle n \rangle) \neq \text{Zero}$ .

2  $(\text{Sub1 (Add1 } \langle n \rangle)) = (\text{Cdr (Cons False } \langle n \rangle)) = \langle n \rangle$ .

Hence:  $(\text{Add1 } \langle n \rangle) = (\text{Add1 } \langle m \rangle) \Rightarrow \langle n \rangle = (\text{Sub1 (Add1 } \langle n \rangle)) = (\text{Sub1 (Add1 } \langle m \rangle)) = \langle m \rangle$ , id est  
 $(\text{Add1 } \langle n \rangle) = (\text{Add1 } \langle m \rangle) \Rightarrow \langle n \rangle = \langle m \rangle$

```
(def Y (λ (m) ((λ (f) (m (f f))) (λ (f) (m (f f))))))
(def + (Y (λ (+ x y) (If (Zero? x) y (If (Zero? y) x (Add1 (Add1 (+ (Sub1 x) (Sub1 y))))))))
(def - (Y (λ (- x y) (If (Zero? x) Zero (If (Zero? y) x (- (Sub1 x) (Sub1 y)))))))
(def * (Y (λ (* x y) (If (Zero? x) Zero (If (Zero? y) Zero (Sub1 (+ (+ (* (Sub1 x) (Sub1 y)) x) y))))))
(def = (Y (λ (= x y) (If (Zero? x) (Zero? y) (If (Zero? y) False (= (Sub1 x) (Sub1 y))))))
(def > (Y (λ (> x y) (If (Zero? x) False (If (Zero? y) True (> (Sub1 x) (Sub1 y))))))
(def < (λ (x y) (> y x)))
(def >= (λ (x y) (Not (< x y))))
(def <= (λ (x y) (Not (> x y))))
(def Quotient (Y (λ (Quotient x y) (If (< x y) Zero (Add1 (Quotient (- x y) y)))))
(def Modulo (λ (x y) (- x (* (Quotient x y) y))))
(def Divisor? (λ (x y) (Zero? (Modulo x y))))

(def Even?+Odd? ; a pair of mutually recursive functions.
  (Y
    (λ (Even?+Odd?)
      ((λ (Even? Odd?)
        (Cons (λ (n) (If (Zero? n) True (Odd? (Sub1 n))))
              (λ (n) (If (Zero? n) False (Even? (Sub1 n))))))
        (λ (n) (Car Even?+Odd? n))
        (λ (n) (Cdr Even?+Odd? n))))))
```

```
(def Even? (Car Even?+Odd?))      (def Odd? (Cdr Even?+Odd?))

; The following two functions provide conversion between natural numbers of Scheme
; and numerals represented by pairs.

(define (number->pair-numeral n)
  (if (zero? n) (ev Zero)
      ((ev (Cons False)) (number->pair-numeral (sub1 n)))))

(define (pair-numeral->number numeral)
  (if (eq? ((ev (λ (numeral) (If (Zero? numeral) yes no))) numeral) 'yes) 0
      (add1 (pair-numeral->number ((ev Sub1) numeral)))))

(provide number->pair-numeral pair-numeral->number)
```

```
#lang scheme ; File test-pair-numerals.ss
(require redex "pair-numerals.ss" "lazy-evaluator.ss" redex)

(for ((n (in-range 0 25)))
  (test-equal (pair-numeral->number (number->pair-numeral n)) n))

(for ((m (in-range 1 11)) #:when #t (n (in-range 1 11)))
  (for-each
    (λ (function check-function)
      (test-equal
        (pair-numeral->number
          ((function (number->pair-numeral m)) (number->pair-numeral n)))
        (check-function m n)))
    (list (ev +) (ev -) (ev *) (ev Modulo) (ev Quotient))
    (list + (λ (m n) (max 0 (- m n))) * modulo quotient)))

(for ((n (in-range 0 10)))
  (test-equal (((ev Even?) (number->pair-numeral n)) #t) #f) (even? n))
  (test-equal (((ev Odd?) (number->pair-numeral n)) #t) #f) (odd? n))

(test-results)
```

```
#lang scheme ; File primes.ss
(require "lazy-evaluator.ss" "pair-numerals.ss" redex)

(def Naturals ((Y (λ (make-naturals n) (Cons n (make-naturals (Add1 n))))) Zero))

(def filter
  (Y
    (λ (filter pred list)
      (If (pred (Car list)) (Cons (Car list) (filter pred (Cdr list))) (filter pred (Cdr list))))))

(def Primes
  ((Y
    (λ (make-primes number-list)
      (Cons (Car number-list)
        (make-primes
          (filter (λ (n) (Not (Divisor? n (Car number-list)))) (Cdr number-list))))))
    (Cdr (Cdr Naturals))))
```

```

(define (list-numbers numbers n)
  (if (zero? n) '()
      (cons (pair-numeral->number ((ev Car) numbers))
            (list-numbers ((ev Cdr) numbers) (sub1 n)))))

(test-equal (list-numbers (ev Naturals) 20)
  '(0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19))

(define (primes n)
  (let loop ((primes '()) (n 2) (k n))
    (cond
      ((zero? k) (reverse primes))
      ((ormap (λ (p) (zero? (modulo n p))) primes) (loop primes (add1 n) k))
      (else (loop (cons n primes) (add1 n) (sub1 k))))))

(test-equal (list-numbers (ev Primes) 25) (primes 25))
(test-results)

```

### Church numerals

$n$  represented by  $[n] \equiv (\lambda (f) (\lambda (x) (f^n x)))$ , where  $(f^0 x) \equiv x$  and  $(f^{n+1} x) \equiv (f (f^n x))$ . Hence a Church numeral is a function that given a function  $f$  returns a function  $f^n$  of argument  $x$  that applies  $f$   $n$  times to argument  $x$ . For example:

```

[0] ≡ (λ (f) (λ (x) x))
[1] ≡ (λ (f) (λ (x) (f x)))
[2] ≡ (λ (f) (λ (x) (f (f x))))
[3] ≡ (λ (f) (λ (x) (f (f (f x)))))
[4] ≡ (λ (f) (λ (x) (f (f (f (f x))))))
[5] ≡ (λ (f) (λ (x) (f (f (f (f (f x)))))))

```

Notice that  $(f (f^n x)) \equiv (f^n (f x))$  and more generally  $(f^n (f^m x)) \equiv (f^{n+m} x)$ . Also notice that in file `church-numerals.ss` no  $Y$  combinator is used for the preparation of the basic numerical functions.

```

#lang scheme ; File church-numerals.ss
(require redex "lazy-evaluator.ss" "booleans.ss")

(define-syntax Church-numeral->number
  (syntax-rules ()
    ((_ numeral) (((ev numeral) (λ (n) (add1 (force n)))) 0))))

(define-syntax test-Church
  (syntax-rules ()
    ((_ numeral expect)
      (test-equal (((ev numeral) (λ (n) (add1 (force n)))) 0) expect))))

(def C-add1 (λ (n f x) (f (n f x))))
(def C-sub1 (λ (n f x) (n (λ (g h) (h (g f))) (λ (f) x) (λ (f) f))))
(def C-plus (λ (m n f x) (m f (n f x))))
(def C-mult (λ (m n f) (n (m f))))
(def C-expt (λ (m n) (n m)))
(def C-zero? (λ (n) (n (λ (x) false) True)))
(def C-minus (λ (m n) (n C-sub1 m)))
(def C-eq? (λ (m n) (And (C-zero? (C-minus m n)) (C-zero? (C-minus n m)))))
(def C0 (λ (f x) x))
(def C1 (λ (f x) (f x)))
(def C2 (λ (f x) (f (f x))))
(def C3 (λ (f x) (f (f (f x)))))

```

```

(def C4 (λ (f x) (f (f (f (f x))))))
(def C5 (λ (f x) (f (f (f (f (f x)))))))
(def C6 (λ (f x) (f (f (f (f (f (f x))))))))
(def C7 (λ (f x) (f (f (f (f (f (f (f x)))))))))
(def C8 (λ (f x) (f (f (f (f (f (f (f (f x))))))))))
(def C9 (λ (f x) (f (f (f (f (f (f (f (f (f x)))))))))))

(test-Church C0 0)
(test-Church C1 1)
(test-Church C2 2)
(test-Church C3 3)
(test-Church C4 4)
(test-Church C5 5)
(test-Church C6 6)
(test-Church C7 7)
(test-Church C8 8)
(test-Church C9 9)
(test-Church (C-add1 C7) 8)
(test-Church (C-sub1 C7) 6)
(test-Church (C-sub1 C0) 0)
(test-Church (C-plus C5 C7) 12)
(test-Church (C-mult C5 C7) 35)
(test-Church (C-minus C5 C5) 0)
(test-Church (C-minus C7 C5) 2)
(test-Church (C-expt C7 C3) 343)
(test-Church (C-expt C5 C4) 625)
(test-equal (ev (If (C-eq? C7 C5) yes no)) 'no)
(test-equal (ev (If (C-eq? C5 C7) yes no)) 'no)
(test-equal (ev (If (C-eq? C0 C0) yes no)) 'yes)
(test-equal (ev (If (C-eq? C1 C1) yes no)) 'yes)
(test-equal (ev (If (C-eq? C7 C7) yes no)) 'yes)
(test-equal (ev (If (C-zero? (C-minus C5 C5)) yes no)) 'yes)
(test-equal (ev (If (C-zero? (C-minus C5 C7)) yes no)) 'yes)
(test-equal (ev (If (C-zero? (C-minus C5 C4)) yes no)) 'no)

(test-results) ; Displays: All 28 tests passed.

```

## Combinatory logic

Consider the following combinators (lambda terms without free variables):

$$\begin{aligned} \mathbf{I} &\equiv (\lambda (x) x) \\ \mathbf{K} &\equiv (\lambda (x y) x) \\ \mathbf{S} &\equiv (\lambda (x y z) ((x z) (y z))) \\ \mathbf{X} &\equiv (\lambda (x) (x \mathbf{K} \mathbf{S} \mathbf{K})) \end{aligned}$$

Now we have:

$$\begin{aligned} (\mathbf{K} A B) &= A \\ (\mathbf{S} A B C) &= ((A C) (A C)) \\ (\mathbf{X} A) &= (A \mathbf{K} \mathbf{S} \mathbf{K}) \end{aligned}$$

Now:

$$\begin{aligned} (\mathbf{X} \mathbf{X} \mathbf{X}) &= ((\mathbf{X} \mathbf{K} \mathbf{S} \mathbf{K}) \mathbf{X}) = (((\mathbf{K} \mathbf{K} \mathbf{S} \mathbf{K}) \mathbf{S} \mathbf{K}) \mathbf{X}) = (((\mathbf{K} \mathbf{K}) \mathbf{S} \mathbf{K}) \mathbf{X}) = (\mathbf{K} \mathbf{K} \mathbf{X}) = \mathbf{K} \\ (\mathbf{X} (\mathbf{X} \mathbf{X})) &= (\mathbf{X} \mathbf{X} \mathbf{K} \mathbf{S} \mathbf{K}) = (\mathbf{X} \mathbf{K} \mathbf{S} \mathbf{K} \mathbf{K} \mathbf{S} \mathbf{K}) = (\mathbf{K} \mathbf{K} \mathbf{S} \mathbf{K} \mathbf{S} \mathbf{K} \mathbf{K} \mathbf{S} \mathbf{K}) = (\mathbf{K} \mathbf{K} \mathbf{S} \mathbf{K} \mathbf{K} \mathbf{S} \mathbf{K}) = (\mathbf{K} \mathbf{S} \mathbf{K}) = \mathbf{S} \end{aligned}$$

$(\mathbf{s} \ \mathbf{k} \ A) = (\lambda (\langle \text{var} \rangle) (\mathbf{s} \ \mathbf{k} \ A \ \langle \text{var} \rangle)) = (\lambda (\langle \text{var} \rangle) ((\mathbf{k} \ \langle \text{var} \rangle) (A \ \langle \text{var} \rangle))) = (\lambda (\langle \text{var} \rangle) \langle \text{var} \rangle) \equiv \mathbf{i}$ , for arbitrary term  $A$  and assuming that  $\langle \text{var} \rangle$  is chosen such as having no free occurrence in  $A$ . Hence:

$\mathbf{i} = (\mathbf{s} \ \mathbf{k} \ \mathbf{k}) = ((\mathbf{x} \ (\mathbf{x} \ \mathbf{x})) (\mathbf{x} \ \mathbf{x} \ \mathbf{x}) (\mathbf{x} \ \mathbf{x} \ \mathbf{x}))$

$\mathbf{k} = (\mathbf{x} \ \mathbf{x} \ \mathbf{x})$

$\mathbf{s} = (\mathbf{x} \ (\mathbf{x} \ \mathbf{x}))$

Hence we only need  $\mathbf{x}$  in order to make  $\mathbf{k}$ ,  $\mathbf{s}$  and  $\mathbf{i}$  as well.  $\{\mathbf{x}\}$  is a one point basis for the set of all lambda terms modulo equality. There are more terms like  $\mathbf{x}$  that form a one point basis. The following scheme shows how abstractions can be eliminated from every term.

$(\lambda (\langle \text{var} \rangle_0) \langle \text{var} \rangle_0)) = \mathbf{i}$

$(\lambda (\langle \text{var} \rangle_0) \langle \text{term} \rangle) = (\mathbf{k} \ \langle \text{term} \rangle)$  if  $\langle \text{var} \rangle_0$  has no free occurrence in  $\langle \text{term} \rangle$

$(\lambda (\langle \text{var} \rangle) (\langle \text{term} \rangle_0 \ \langle \text{term} \rangle_1)) = (\mathbf{s} \ (\lambda (\langle \text{var} \rangle) (\langle \text{term} \rangle_0 \ \langle \text{var} \rangle)) (\lambda (\langle \text{var} \rangle) (\langle \text{term} \rangle_1 \ \langle \text{var} \rangle)))$

By repeating these rules on a given term, an equal term is obtained in terms of  $\mathbf{i}$ ,  $\mathbf{k}$  and  $\mathbf{s}$ , free variables and application. Furthermore  $\mathbf{i}$ ,  $\mathbf{k}$  and  $\mathbf{s}$  can be expressed in  $\mathbf{x}$  such as to obtain a term with  $\mathbf{x}$ , free variables and application only. The transformation is shown in the following program:

```
#lang scheme ; File one-point-basis.ss
(require redex "curry.ss" "curried-lc-grammar.ss" "free-vars.ss")

(define-metafunction curried-lc-grammar Trafo : <term> -> <term>
  ((Trafo I) (Trafo ((S K) K)))
  ((Trafo K) ((X X) X))
  ((Trafo S) (X (X X)))
  ((Trafo X) X)
  ((Trafo <var>) <var>)
  ((Trafo (λ (<var>_0) <var>_0)) (Trafo I))
  ((Trafo
    (side-condition
      (λ (<var>_0) <term>_0)
      (not (term (Var-free-in? <var>_0 <term>_0))))))
    (Trafo K) (Trafo <term>_0)))
  ((Trafo (λ (<var>) (<term>_0 <term>_1)))
    (((Trafo S) (Trafo (λ (<var>) <term>_0))) (Trafo (λ (<var>) <term>_1))))
  ((Trafo (λ (<var>) <term>)) (Trafo (λ (<var>) (Trafo <term>))))
  ((Trafo (<term>_0 <term>_1)) ((Trafo <term>_0) (Trafo <term>_1))))

(define K (term (Curry (λ (x y) x))))
(define S (term (Curry (λ (x y z) ((x z) (y z))))))
(define X (term (Curry (λ (x) (x , K , S , K)))))

(define ns (make-base-namespace))
(define (ev x) (eval x ns))
(ev '(require "free-vars-in-scheme.ss"))

(define-syntax test
  (syntax-rules ()
    ((_ p q)
      (let* ((x (term (Trafo (Curry p)))) (x-term (term ((λ (X) , x) , X))))
        (printf "~s ->~n" 'x)
        (pretty-display x)
        (printf "-> ~s~n~n" 'q)
        (test-equal (ev x-term) 'q)))))

(test ((λ (x) x) yes) yes)
```



```

(test ((λ (x y) x) yes no) yes)
(test ((λ (x y) y) yes no) no)
(test (X X a b c) b)
(test (S K X (S K K) b) b)
(test (S K K b) b)
(test (S K S b) b)
(test (K yes no) yes)
(test (K I yes no) no)
(test (K I I yes) yes)
(test (K I I I I yes) yes)
(test (S I I K a b c) b)
(test (S S S S S S I I I yes) yes)
(test (S (S S) (S S) (S S) S S K yes) yes)
(test (X X X X X X I I yes) yes)
(test (X (X X) (X X) (X X) X X I I yes) yes)
(test-results)

```

## Listing all combinators modulo equality

Because every combinator (term without free variables) can be expressed in **x** with application but without abstractions and without free variables, it is possible to prepare a combinator that lists all combinators modulo equality. The list is infinite of course. A finite solution is a combinator **E** such that  $(\mathbf{E} [n])$  is the  $n^{\text{th}}$  term of the list. The list certainly has duplicates. It is very well possible that  $(\mathbf{E} [n]) = (\mathbf{E} [m])$  for some  $n \neq m$ . The construction of term **E** is not shown here (See [Barendregt](#)) Below Scheme and *redex* are used. Let the number of occurrences of **x** be the length of an X-term. We make a function that lists all X-terms with length n. Concatenation of all these lists results in the infinite list of all combinators.

```

#lang scheme ; file term-generator.ss
(require redex)

(define-language x-lang
  (<term> X (<term> <term>))
  (<hole> (<hole> <term>) (<term> <hole>) hole))

(define extend
  (reduction-relation x-lang (--> X (X X))
    (--> (in-hole <hole> X) (in-hole <hole> (X X)))))

(define (list-terms n) ; Lists all terms with n occurrences of x.
  (case n
    ((0) '())
    ((1) '(X))
    (else (remove-duplicates (apply append (map apply-extend (list-terms (sub1 n)))))))

(define-metafunction x-lang Term<? : <term> <term> -> any
  ((Term<? X X) #f)
  ((Term<? X <term>) #t)
  ((Term<? <term> X) #f)
  ((Term<? (<term>_0 <term>_1) (<term>_0 <term>_2)) (Term<? <term>_1 <term>_2))
  ((Term<? (<term>_0 <term>_1) (<term>_2 <term>_3)) (Term<? <term>_0 <term>_2)))

(define (Catalan n) (quotient (! (* 2 n)) (* (! n) (! (add1 n))))) ; (2n)/(n!(n+1)!)
; These are Catalan numbers. They solve many counting problems.
; Here they are used to count the number of fully curried X-terms with n+1 occurrences of x, n ≥ 0.

```

```
(define (sort-terms x) (sort x term<?))
(define (term<? x y) (term (Term<? , x , y)))
(define (apply-extend x) (apply-reduction-relation extend x))
(define (! n) (if (zero? n) 1 (* n (! (sub1 n)))))
(define print-terms #f) ; Replace #f by #t if you want to see the lists of terms.

(for ((n (in-range 1 10)))
  (let* ((r (sort (list-terms n) term<?)) (c (length r)) (C (Catalan (sub1 n))))
    (printf "nr of X: ~s, nr of terms: ~s~n" n c)
    (when print-terms (for-each (λ (r) (printf "~s~n" r)) r) (newline))
    (test-equal c C)))

(test-results)
```

## Results:

```
nr of X: 1, nr of terms: 1
nr of X: 2, nr of terms: 1
nr of X: 3, nr of terms: 2
nr of X: 4, nr of terms: 5
nr of X: 5, nr of terms: 14
nr of X: 6, nr of terms: 42
nr of X: 7, nr of terms: 132
nr of X: 8, nr of terms: 429
nr of X: 9, nr of terms: 1430
All 9 tests passed.
```

## Appendix A

If you don't yet have DrScheme, download it from [the PLT Scheme download page](#), execute the downloaded installer and click 'next' until the installer starts. DrScheme is free. After the installer completes click 'finish' and DrScheme opens. Select 'Language Module' in menu item 'Language/Choose Language'. Also select menu item 'Edit/Preferences/General' and enable 'Automatically switch to module language when opening a module'. Personally I also prefer the options 'Open files in separate tabs' and 'Put interactions window beside the definitions window'. It may be handy to enable the option 'Automatically compile source files' in menu item 'Language/Choose Language'. Show details when hidden in order to get the option visible. Setting the language and the preferences is a one time job. DrScheme memorizes the settings. If you are not yet familiar with DrScheme, you may want to have a look at [DrScheme: PLT Programming Environment](#), although in principle, the user interface hardly requires any explanation. In the present essay frequent use is made of PLT's [redex](#). It has a [beginner friendly introduction](#), but the latter requires some knowledge of Lambda Calculus. In the present essay the first examples referring to *redex* are self evident. Take a look at the introduction when you feel you are up to it. I make frequent use of [Check Syntax](#). It provides a lot of information about the structure of the program. Many typo errors can be traced by means of Check Syntax.

The tools and source codes are available as a zip file: [lc-with-redex.zip](#). Download it and unzip. All files must be kept in the same directory. The directory contains the original copy of the present essay. Always use the copy in the directory. Many hyperlinks in this essay assume the referenced material to be present in the same directory as the essay itself. Run examples by following a hyperlink in this essay or by opening one of the '.ss' files in the directory. Run DrScheme.exe for your own programs. Enter your program in the definitions window. Results are shown in the interactions window after clicking the Run button and in some cases a little bit of patience. The interactions window also provides a read-eval-print-loop, but I rarely use it.

## Copyright

You can use this essay and the sources in directory [lc-with-redex.zip](#) as you like, but if do you this in public, it would be nice to refer to the origin: Jacob J. A. Koot or less formally Jos Koot. I must demand, though, that you make sure that when referring to parts that I borrowed from others, the related acknowledgments are maintained.